
Plan du cours 2

Polymorphisme paramétrique:
▶ λ-calcul typé

▶ types simples
▶ vérification d’un type
▶ unification
▶ synthèse de types

▶ schémas de types et polymorphisme

Préambule
On a vu que le λ-calcul est la pure théorie de la fonctionalité. Mais
on peut vouloir associer à chaque variable d’un terme un type qui
indique sa nature fonctionnelle, c’est-à-dire ses domaine et
codomaine. C’est naturel quand on pense aux mathématiques où on
s’intéresse rarement à une fonction indépendamment de son
ensemble de départ et de son ensemble d’arrivée. Les types
permettent de retrouver cette idée dans le λ-calcul.

De plus, ils ont un intérêt calculatoire, car l’intérêt d’un système de
types pour le λ-calcul (d’un point de vue théorique) est d’assurer la
forte normalisation des termes typables. Ce sera le cas du système
que nous allons voir, qui est le plus simple de tous. C’est pourquoi
on l’appelle le système des types simples. Dans cette optique, le
λ-calcul est vu comme un langage de programmation, et les types
assurent la terminaison des programmes typables (indépendamment
de la stratégie de réduction).

Types simples

La syntaxe du λ-calcul est identique à celle qui a été introduite
précédemment.

Soit P un ensemble de variables de types. On définit l’ensemble des
types simples bien formés T (construit à partir de T et du
constructeur →) par :
▶ si α ∈ P alors α ∈ T
▶ si σ, τ ∈ T alors σ → τ ∈ T .

exemples:
(α → α) → (α → α)
(α → β) → α → β

Contexte de typage

Typer un terme (clos), c’est lui associer un type. Mais on ne peut
pas se contenter de ne typer que les termes clos. Il faut donc se
référer à un contexte (ou environnement) de types (qui associe à
chaque variable libre un type).

Donc typer un terme (non nécessairement clos) consistera à lui
associer un type dans un certain contexte où se trouvent toutes les
variables libres du terme. Donnons une définition plus formelle.

Les contextes sont des listes de couples C = (x1 : σ1), . . . , (xn : σn).

typer un terme (1)
On définit inductivement une relation C ⊢ M : τ qui signifie :

“M est de type τ dans le contexte C ”
de la façon suivante :

▶ Si C = (x1 : σ1), . . . , (xn : σn), alors C ⊢ xi : σi pour tout
i = 1, . . . , n (Var).

▶ Si C ⊢ M : σ → τ et C ⊢ N : σ alors C ⊢ MN : τ (App).

▶ Et si (x : σ)C ⊢ M : τ alors C ⊢ λx .M : σ → τ (Abs).

On parle alors de jugement de typage, noté C ⊢ M : τ , qui indique
que sous le contexte C , le terme M est typable avec le type τ .

Le parenthésage du constructeur de type fonctionnel (→) est à
droite : α → β → α ≡ α → (β → α)

typer un terme (2)
Ces règles s’écrivent habituellement :

(Var)
(x1 : σ1), . . . , (xn : σn) ⊢ xi : σi

(App)
C ⊢ M : σ → τ C ⊢ N : σ

C ⊢ MN : τ
(Abs)

(x : σ),C ⊢ M : τ

C ⊢ λx .M : σ → τ
▶ typer une variable : c’est vérifier qu’elle apparait dans le

contexte.
▶ typer une abstraction : lui associer un type τ → σ si le

paramètre a un type τ et le corps un type σ.
▶ typer une application : M doit avoir un type fonctionnel et N

le type du paramètre de M, et MN a le type du résultat.

Exemple de typage (1)

on se propose de typer l’entier de Church “deux”(λf λx .f (fx)). On
prouve qu’il admet le type (α → α) → α → α au moyens des règles
ci-dessus.
1) on a (x : α), (f : α → α) ⊢ x : α

2) et (x : α), (f : α → α) ⊢ f : α → α

3) donc par 1) et 2) (x : α), (f : α → α) ⊢ f x : α

4) par 2) et 3) on a (x : α), (f : α → α) ⊢ f (f x) : α

5) puis par la règle d’abstraction, de 4) on déduit
(f : α → α) ⊢ λx .f (f x) : α → α

6) et de même (f : α → α) ⊢ λf .λx .f (f x) : (α → α) → α → α

Exemple de typage (2)

Ce qui s’écrit plus facilement en utilisant la deuxième forme des
règles :

x : α, f : α→ α ⊢ f : α→ α

x : α, f : α→ α ⊢ f : α→ α x : α, f : α→ α ⊢ x : α

x : α, f : α→ α ⊢ fx : α

x : α, f : α→ α ⊢ f (fx) : α

f : α→ α ⊢ λx .f (fx) : α→ α

⊢ λf .λx .f (fx) : (α→ α)→ α→ α

On a donné une méthode de vérification de typage (lecture de haut
en bas), mais pas de synthèse (création de l’arbre de bas en haut)
d’un type.

Cours 2 - Typage et Analyse Statique (TAS - MU5IN555) - année 2023/2024 - 9 / 28



Synthèse (ou inférence) (1)
On a donné une méthode de vérification de typage, mais pas de
synthèse d’un type.

Pour ce système de types, il existe une méthode de synthèse
utilisant l’unification, que nous allons donner.

En voici un avant-goût par le raisonnement intuitif suivant : On
suppose que dans ce terme, f a le type σ et x a le type τ .

Comme on a pu former f x il faut que σ soit de la forme τ → σ1 et
alors f x est du type σ1. Comme on a pu former f (f x), il faut que
τ = σ1. En conclusion, f : τ → τ et x : τ , et le terme a le type
(τ → τ) → τ → τ pour n’importe quel type τ (en particulier, on
peut prendre une variable α).

L’idée est de poser les contraintes de typage en remontant la
dérivation, puis de les résoudre par unification. Si échec le terme
n’est pas typable.

Synthèse (ou inférence) (2)
En partant du terme à typer , on crée des contraintes de typage à
chaque règle utilisée dans l’arbre de dérivation, puis on unifie ces
contraintes. Si l’unification réussit on récupère le type le plus
général, et sinon cela indique que le terme n’est pas typable.

avec K = λz1z2.z1, I1 = λy .y et I2 = λx .x , on cherche à typer
K I1 I2 on construit l’arbre de dérivation :

z1 : t4, z2 : t1 ⊢ z1 : t0
z1 : t4 ⊢ λz2.z1 : t1 → t0
⊢ K : t4 → t1 → t0

y : t5 ⊢ y : t6
⊢ I1 : t4

⊢ K I1 : t1 → t0

x : t2 ⊢ x : t3
⊢ I2 : t1

⊢ K I1 I2 : t0

et on récupère cette liste de contraintes de typage :
{t1 = t2 → t3; t2 = t3; t4 = t5 → t6; t5 = t6; t0 = t4}
qui s’unifie en :
{t0 = t4 = α → α; t1 = β → β; t2 = t3 = α; t5 = t6 = β}

Synthèse (ou inférence) (3)

Termes non typables:

échec de l’unification sur le typage du terme λx .x x :
▶ la variable x doit être d’un type α et en même temps d’un

type α → β car x s’applique sur x .

Synthèse (ou inférence) (4)

L’exploration du terme construit la dérivation avec :
▶ les règles (Var) produisent des contraintes sur les variables de

type
▶ les règles Abs produisent des contraintes et engendre des

variables
▶ les règles App engendrent des variables de type

On va s’intéresser à l’algorithme d’Hindley-Milner pour un λ-calcul
simplement typé.

Synthèse par unification
On va rappeler l’algorithme d’unification, dans le cas simple qui est
le nôtre. On considère une signature ne comportant qu’un seul
symbole, une fonction a d’arité 2 (c’est la flèche → des types). Les
variables que nous considérerons sont celles de l’ensemble P.
Rappelons qu’une substitution est une application P → T (T est
l’ensemble des termes bâtis sur cette signature), et qu’on peut
appliquer une substitution s à un terme σ ∈ T , ce qui donne un
terme σs :
▶ xs = s(x)

▶ (a(σ, τ))s = a(σs, τs)

Cela permet de définir la composée s ◦ t de deux substitutions
comme une substitution.
Le problème consiste à résoudre des équations dans cette signature,
et de trouver la solution la plus générale possible. Pour donner un
sens à cela, il faut mettre un ordre sur les substitutions.

Unification (1)

Définition
On dit que s est plus général que t s’il existe une substitution u
telle que u ◦ s = t.
Si σ, τ ∈ T , unifier σ et τ , c’est trouver une substitution s telle que
σs = τs. Alors le résultat est le suivant :

Théorème
Pour tous σ, τ ∈ T unifiable, il existe un unificateur qui est plus
général que tous les autres. On l’appelle unificateur principal, et on
le note UP(σ, τ).
La preuve de ce théorème ne présente d’intérêt que par l’algorithme
d’unification qu’elle propose, qui calcule cet unificateur principal ou
échoue. Nous rappelons à présent cet algorithme. On remarquera
qu’ici la seule cause d’échec possible est celle qui résulte du test
d’occurrence d’une variable.

Unification (2)
On notera [σ/α] la substitution qui envoie α sur σ et toutes les
autres variables sur elles-mêmes. Alors, on définit la fonction UP
qui prend deux termes et rend une substitution inductivement par :
▶ UP(α, σ) = [σ/α] si α ne figure pas dans σ, et échec si non.
▶ UP(σ, α) = [σ/α] si α ne figure pas dans σ, et échec si non.
▶ si s = UP(σ, σ′) et t = UP(τs, τ ′s), alors

UP(a(σ, τ), a(σ′, τ ′) = t ◦ s).
Justification de l’algorithme (ce n’est pas une preuve rigoureuse).
Le seul cas non trivial est celui où l’on a à unifier a(σ1, τ1) avec
a(σ2, τ2). Alors si s = UP(σ1, σ2), alors σ1s = σ2s et si
t = UP(τ1s, τ2s), alors (τ1s)t = (τ2s)t, c’est-à-dire τ1u = τ2u où
u = t ◦ s. On voit facilement que u est un unificateur de nos deux
termes de départ. Il est principal car sinon il existerait une
substitution, non contenue dans (t; s), qui composée à lui
atteindrait l’UP du terme, ce qui est contradictoire avec sa
construction.

Implantation
Voici l’algorithme d’unification dans le cas des types simples. La
syntaxe abstraite des types :

1 type oType =

2 Tvar of string

3 | Arrow of oType * oType;;

On représente les substitutions par des listes de couples chaîne de
caractère, terme, Voici l’application d’une substitution à un type :

1 (* val substitute : (string * oType) list -> oType -> oType = <fun> *)

2

3 let substitute s t =

4 let rec sub_rec t = match t with

5 (Tvar a) as alpha -> (try List.assoc a s with Not_found -> alpha)

6 | Arrow(oT1,oT2) -> Arrow(sub_rec oT1, sub_rec oT2)

7 in sub_rec t

8 ;;

et voici la composition de deux substitutions :

1 (* val compsubst : (string * oType) list -> (string * oType) list -> (←↩
string * oType) list = <fun> *)

2

3 let compsubst s t =

4 (List.map (function (a,oT) -> (a,substitute s oT)) t) @ s;;

La fonction occur teste l’existence d’une variable dans un type.

1 (* val occur : string -> oType -> bool = <fun> *)

2

3 let occur v t =

4 let rec occ_rec t = match t with

5 Tvar b -> b=v

6 | Arrow(oT1,oT2) -> occ_rec oT1 or occ_rec oT2

7 in occ_rec t

8 ;;

La fonction unify calcule l’unificateur principal pour deux types
simples. Elle retourne cet unificateur quand il existe ou échoue en
déclenchant l’exception Failure "unify".

1 (* val unify : oType * oType -> (string * oType) list = <fun> *)

2

3 let rec unify c = match c with

4 (Tvar a), oT ->

5 if oT=Tvar a then []

6 else if occur a oT then raise (Failure "unify")

7 else [a,oT]

8 | oT,Tvar a ->

9 if occur a oT then raise (Failure "unify")

10 else [a,oT]

11 | Arrow(oS1,oS2), Arrow(oT1,oT2) ->

12 let s=unify(oS1,oT1) in

13 compsubst (unify(substitute s oS2,

14 substitute s oT2)) s

15 ;;

Les exemples suivants calculent l’unificateur principal de deux
types. Cette substitution est ensuite appliquée aux types pour
vérifier leur égalité après substitution.

1. Un des cas les plus simples où une variable de type est
substituée par une autre :

1 # let oT1=(Tvar "x");;

2 val oT1 : oType = Tvar "x"

3 # let oT2=(Tvar "y");;

4 val oT2 : oType = Tvar "y"

5 # let s1 = unify (oT1,oT2);;

6 val s1 : (string * oType) list = [("x", Tvar "y")]

7 # substitute s1 oT1 = substitute s1 oT2;;

8 - : bool = true

2. Un cas classique où une variable est substituée par un type
fonctionnel :

1 # let oT3 = (Arrow (Tvar "a", Arrow (Tvar "b",Tvar "c")));;

2 val oT3 : oType = Arrow (Tvar "a", Arrow (Tvar "b", Tvar "c"))

3 # let oT4 = (Arrow (Arrow (Tvar "d",Tvar "e"), (Arrow (Tvar "f",Tvar "g←↩
"))));;

4 val oT4 : oType =

5 Arrow (Arrow (Tvar "d", Tvar "e"), Arrow (Tvar "f", Tvar "g"))

6 # let s2 = unify (oT3,oT4);;

7 val s2 : (string * oType) list =

8 [("a", Arrow (Tvar "d", Tvar "e")); ("b", Tvar "f"); ("c", Tvar "g")]

9 # substitute s2 oT3;;

10 - : oType = Arrow (Arrow (Tvar "d", Tvar "e"), Arrow (Tvar "f", Tvar "g←↩
"))

11 # substitute s2 oT4;;

12 - : oType = Arrow (Arrow (Tvar "d", Tvar "e"), Arrow (Tvar "f", Tvar "g←↩
"))

3. Un exemple où l’unification fait disparaitre presque toutes les
variables :

1 # let oT5 = (Arrow (Arrow (Tvar "d",Tvar "a"), (Arrow (Tvar "a",Tvar "g←↩
"))));;

2 val oT5 : oType =

3 Arrow (Arrow (Tvar "d", Tvar "a"), Arrow (Tvar "a", Tvar "g"))

4 # let oT6 = (Arrow (Arrow (Tvar "a",Tvar "e"), (Arrow (Tvar "a",Tvar "d←↩
"))));;

5 val oT6 : oType =

6 Arrow (Arrow (Tvar "a", Tvar "e"), Arrow (Tvar "a", Tvar "d"))

7 # let s3 = unify (oT5,oT6);;

8 val s3 : (string * oType) list =

9 [("d", Tvar "e"); ("a", Tvar "e"); ("g", Tvar "e")]

10 # substitute s3 oT5 = substitute s3 oT6;;

11 - : bool = true

4. Enfin un cas d’échec :

1 # unify (oT3,oT4);;

2 - : (string * oType) list =

3 [("a", Arrow (Tvar "d", Tvar "e")); ("b", Tvar "f"); ("c", Tvar "g")]

4 # unify (oT3,oT5);;

5 Exception: Failure "unify".

On peut maintenant donner l’algorithme de typage.

Algorithme de typage
On définit une fonction T qui prend un contexte et un terme M et
rend un couple constitué d’une substitution et du type cherché pour
M dans ce contexte (on produit le type le plus général possible).
▶ T (x ,C ) = (C (x), id)
▶ Pour calculer T (λx .M,C ), on introduit une nouvelle variable

de type α, et on essaie de typer M dans le nouveau contexte
(x : α)C ; soit (σ, s) = T (M, (x : α)C ). Alors le type cherché
est s(α) → σ, et on rend la substitution s.

▶ Pour calculer T (MN,C ), on calcule (σ, s) = T (M,C ), et
(τ, t) = T (N,Cs), car il faut tenir compte des contraintes
induites par le typage de M. On veut pouvoir appliquer M à
N, et pour cela il faut que M ait un type fonctionnel dans le
contexte courant, c’est-à-dire que σt soit de la forme τ → ϕ
où ϕ est un type à déterminer. Pour cela, on calcule
l’unificateur principal u de σt et τ → α où α est une nouvelle
variable de type. On rend le type u(α), et la substitution
u ◦ t ◦ s.

Implantation (1)

Voici le programme de typage en Caml :
Le type term pour la représentation des λ-termes et la fonction
gensym pour la création des variables de type sont définis ainsi :

1 type term= Var of string

2 | Abs of string * term

3 | App of term * term;;

1 # let c = ref 0;;

2 val c : int ref = {contents = 0}

3 # let gensym s = (c:=!c+1 ;s^(string_of_int !c));;

4 val gensym : string -> string = <fun>

5 # let reset () = c := 0;;

6 val reset : unit -> unit = <fun>

Implantation (2)
La fonction suivante TYPE suit très exactement l’algorithme de
typage indiqué ci-dessus.

1 (* val oTYPE : (string * oType) list -> term -> oType * (string * oType) ←↩
list = <fun> *)

2

3 let rec oTYPE oC t = match t with

4 Var x -> List.assoc x oC , []

5 | Abs(x,oM) ->

6 let alpha=gensym "a" in

7 let (sigma,s) = oTYPE ((x,Tvar alpha)::oC) oM in

8 Arrow( (try List.assoc alpha s with _ -> Tvar alpha),

9 sigma), s

10 | App(oM,oN) ->

11 let (sigma,s)=oTYPE oC oM in

12 let (tau,t) = oTYPE (List.map (function (x,phi) ->

13 (x,substitute s phi)) oC) oN in

14 let alpha=gensym "a" in

15 let u= unify(substitute t sigma, Arrow(tau,Tvar alpha)) in

16 (try List.assoc alpha u with _ -> Tvar alpha),

17 compsubst u (compsubst t s)

18 ;;

Implantation (3)
La fonction print_type donne un affichage lisible des types.

1 (* val print_type : oType -> unit = <fun> *)

2 let print_type t =

3 let rec ptype t = match t with

4 Tvar x -> print_string x

5 | Arrow (x,y) -> print_string "("; ptype x;print_string " -> ";

6 ptype y;print_string ")"

7 in

8 ptype t;print_newline() ;;

Les exemples suivants reprennent des termes déjà définis.
1. A :

1 # let oA = Abs ( "x" , Abs ("y" , App (Var "x", Var "y") ));;

2 val oA : term = Abs ("x", Abs ("y", App (Var "x", Var "y")))

3 # oTYPE [] oA;;

4 - : oType * (string * oType) list =

5 (Arrow (Arrow (Tvar "a2", Tvar "a3"), Arrow (Tvar "a2", Tvar "a3")),

6 [("a1", Arrow (Tvar "a2", Tvar "a3"))])

7 # print_type (fst (oTYPE [] oA));;

8 ((a5 -> a6) -> (a5 -> a6))

9 - : unit = ()

Implantation (4)
1. S :

1 # let oS = Abs( "x" ,

2 Abs ( "y" ,

3 Abs ( "z", App ( App (Var "x" , Var "z" ) ,

4 App (Var "y",Var "z") ) )));;

5 val oS : term =

6 Abs ("x",

7 Abs ("y", Abs ("z", App (App (Var "x", Var "z"), App (Var "y", Var "←↩
z")))))

8 # print_type (fst (oTYPE [] oS));;

9 ((a9 -> (a11 -> a12)) -> ((a9 -> a11) -> (a9 -> a12)))

10 - : unit = ()

2. ∆ :

1 # let delta = Abs ( "x" , App (Var "x" , Var "x" ));;a

2 val delta : term = Abs ("x", App (Var "x", Var "x"))

3 # print_type (fst (oTYPE [] oS));;

4 ((a17 -> (a19 -> a20)) -> ((a17 -> a19) -> (a17 -> a20)))

5 - : unit = ()

6 # print_type (fst (oTYPE [] delta));;

7 Exception: Failure "unify".

Propriétés (1)

▶ subject reduction :
Si C ⊢ M : τ et M → M ′, alors C ⊢ M ′ : τ (le type d’un calcul
est le type de son résultat)

▶ normalisation forte :
On dit qu’un terme M est typable (avec les types simples) s’il
existe un contexte C et un type σ tels que C ⊢ M : σ. Voici le
théorème :

Théorème
Tout terme typable (avec les types simples) est fortement
normalisable.

On n’en donnera pas la preuve.

Les termes comme Ω ou Y (point fixe) ne sont pas typables
dans le système de types simples.

Propriétés (2)

types habités:

On dit que le type τ est habité s’il existe M tel que C ⊢ M : τ

▶ α → α est habité par λx .x
▶ α → β n’est pas habité
▶ (α → β) → α → β ?
▶ (α → β) → β ?

Indication : utilisation des types des paramètres pour le type
résultat.

thèse Gabriel Scherer : " Which types have a unique inhabitant ?
Focusing on pure program equivalence." (2016)

Cours 2 - Typage et Analyse Statique (TAS - MU5IN555) - année 2023/2024 - 24 / 28



Polymorphisme (1)

Le système des types simples ne donne pas le vrai polymorphisme
malgré son ensemble de variables de type.

En effet le terme
▶ (λf .ff )(λx .x)

n’est pas typable car la variable f prend deux valeurs :
▶ α → α

▶ et (α → α) → α → α.
Ce qui est impossible car une variable ne possède qu’un seul type.

Cours 2 - Typage et Analyse Statique (TAS - MU5IN555) - année 2023/2024 - 25 / 28



Polymorphisme (2)

Pour cela on étend le système de type. Soit
▶ P un ensemble de variables de types,
▶ T l’ensemble des types simples construit à partir de T
▶ et S l’ensemble des schémas de types construit à partir de P

et T .

▶ types simples :
▶ si α ∈ P alors α ∈ T
▶ si σ, τ ∈ T alors σ → τ ∈ T .

▶ schémas de type :
▶ si τ ∈ T alors τ ∈ S
▶ si σ ∈ S, α ∈ P alors ∀α.σ ∈ S

Polymorphisme (3)

On obtient ainsi de nouvelles règles de typage :

(Var)
(x1 : σ1), . . . , (xn : σn) ⊢ xi : τ [τi/αi ] σi = ∀α1, . . . , αn.τ

(App)
C ⊢ M : τ → τ ′ C ⊢ N : τ

C ⊢ MN : τ ′
(Abs)

(x : τ),C ⊢ M : τ ′
C ⊢ λx .M : τ → τ ′

(Let)

C ⊢ N : τ α1, . . . , αn = V (τ)− V (C) (x : ∀α1, . . . , αn.τ),C ⊢ M : τ ′
C ⊢ let x = N in M : τ ′

Polymorphisme (4)

Ce qui nous permet de typer let f = λx .x in ff :

VAR
f : ∀β.β → β ⊢ f : (α→ α)→ α→ α

VAR
f : ∀β.β → β ⊢ f : α→ α

⊢ λx .x : β → β β = V (β → β)− V (∅) f : ∀β.β → β ⊢ ff : α→ α

⊢ let f = λx .x in ff : α→ α

C’est le let qui introduit le polymorphisme.

